// Copyright (C) 2010 BIS. All rights reserved.
//
// Redistribution and use in source and binary forms,
// with or without modification, are permitted
// provided that the following conditions are met:
//
// Redistributions of source code must retain the above copyright notice,
// this list of conditions and the following disclaimer.
// Redistributions in binary form must reproduce the above copyright notice,
// this list of conditions and the following disclaimer in the documentation
// and/or other materials provided with the distribution.
// Neither the name of the European Central Bank
// nor the names of its contributors may be used to endorse or promote products
// derived from this software without specific prior written permission.
//
// THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS
// "AS IS" AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED
// TO, THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A
// PARTICULAR PURPOSE ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT
// HOLDERS OR CONTRIBUTORS BE LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL,
// SPECIAL, EXEMPLARY, OR CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT
// LIMITED TO, PROCUREMENT OF SUBSTITUTE GOODS OR SERVICES; LOSS OF USE,
// DATA, OR PROFITS; OR BUSINESS INTERRUPTION) HOWEVER CAUSED AND ON ANY
// THEORY OF LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY, OR TORT
// (INCLUDING NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE OF
// THIS SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE.
package org.bis.core.util.formatter.series
{
	import eu.ecb.core.util.formatter.series.ISeriesTitleFormatter;
	
	import org.sdmx.model.v2.reporting.dataset.KeyValue;
	import org.sdmx.model.v2.reporting.dataset.TimeseriesKey;

	/**
     * Uses the description (in English) of one or more dimensions as a
     * series title. Includes an optional delimiter for the dimension values.
     * 
     * @author Steven Bagshaw 
     * @author Xavier Sosnovsky
     */
	public class MultiDimensionsSeriesTitleFormatter
		implements ISeriesTitleFormatter
	{
		/*==============================Fields================================*/
		
		private var _dimensionCodes:Array;
		private var _delimiter:String = " ";
		
		/*===========================Constructor==============================*/
		
		public function MultiDimensionsSeriesTitleFormatter()
		{
			super();
		}
		
		/*============================Accessors===============================*/
		
		/**
		 * The id of the dimensions (e.g.: REF_AREA, FREQ, etc) to be used for
		 * the series title.
		 */ 
		public function set dimensions(codes:Array):void
		{
			if (codes.length < 1) {
			    throw new ArgumentError("At least one dimension should be" + 
			    	" specified in the series title array.");
			}
			_dimensionCodes = codes;
		}
		
		/**
		 * Specify a String that will appear between dimension labels.
		 */ 
		public function set delimiter(delimiter:String):void
		{
			_delimiter = delimiter;
		}

		/*==========================Public methods============================*/
		
		/**
		 * @inheritDoc
		 */ 
		public function getSeriesTitle(series:TimeseriesKey):String
		{
			if (null == _dimensionCodes) {
				throw new ArgumentError("The dimension(s) from which the title" 
					+ " is generated have not been set");
			}
			
			var title:String = "";
			var firstDimensionDone:Boolean = false;
			
			for each (var dimensionID:String in _dimensionCodes)
			{
				//for now, we'll ignore a specified dimension that does not
				//exist in the series
				for each (var dimension:KeyValue in series.keyValues)
				{
				    if (dimension.valueFor.conceptIdentity.id == dimensionID) {
					    if (firstDimensionDone) {
					        title += _delimiter;
					    }
					    title += dimension.value.description.localisedStrings.
					    	getDescriptionByLocale("en");
					    firstDimensionDone = true;
					    break;
				    }  
				}
			}
			
			if (title.length < 1) {
				throw new ArgumentError("Title could not be found");
			}
			
			return title;
		}
	}
}